Scilabによる古典制御計算

吉 田 和 信

2008年5月11日
2010年9月10日　改訂

目 次

第 1 章 Scilab による古典制御計算 **1**

1.1 多項式の入力，代数方程式の根 1
1.2 伝達関数の入力 1
1.3 伝達関数の結合 2
1.4 時間応答 2
1.5 周波数応答 4
1.6 開ループ伝達関数の特性値 6
1.7 閉ループ伝達関数の特性値 7
1.8 ナイキスト線図 8
1.9 根軌跡 8

第1章　Scilabによる古典制御計算

Scilab ver.5.2.2 を使用しました.

1.1　多項式の入力，代数方程式の根

```
-->s=poly(0,'s');              // 変数 s の定義 （//から右はコメント）
-->s=%s;                       // s には簡略形がある.
-->p1=s^4+2*s^3+3*s^2+4*s+5   // 多項式の入力
 p1  =
               2    3   4
   5 + 4s + 3s + 2s + s
-->roots(p1)                   // p1=0 の根
 ans  =
   0.2878155 + 1.4160931i
   0.2878155 - 1.4160931i
  - 1.2878155 + 0.8578968i
  - 1.2878155 - 0.8578968i
-->p2=(s+1)*(s+2)*(s+3)
 p2  =
                2   3
   6 + 11s + 6s + s
-->roots(p2)
 ans  =
  - 1.
  - 2.
  - 3.
```

1.2　伝達関数の入力

```
-->s=%s;
-->g1=(s+1)/(s^2+2*s+4)
 g1  =
     1 + s
   ---------
            2
   4 + 2s + s
-->g2=(s+2)*(s+3)/(s*(s+1))
 g2  =
            2
   6 + 5s + s
```

```
    ----------
           2
      s + s
-->g2.num      // g2 の分子
 ans  =
              2
    6 + 5s + s
-->g2.den      // g2 の分母
 ans  =
         2
    s + s
```

1.3　伝達関数の結合

```
-->s=%s;
-->g1=1/(s+1);
-->g2=5/(s+2);
-->g3=g1*g2       // 直列結合
 g3  =
        5
    ---------
              2
    2 + 3s + s
-->g4=g1+g2       // 並列結合
 g4  =
      7 + 6s
    ----------
              2
    2 + 3s + s
-->g5=g1/.g2       // フィードバック結合（ネガティブフィードバック）
 g5  =
      2 + s
    ---------
              2
    7 + 3s + s
-->g6=g1/(1+g1*g2)       // フィードバック結合（計算による確認）
 g6  =
      2 + s
    ---------
              2
    7 + 3s + s
```

1.4　時間応答

```
-->s=%s;
-->g1 = 1/(s^2+s+1)
```

```
 g1  =
 
        1
    ---------
             2
    1 + s + s
-->s1=syslin('c',g1)        // 線形系の定義．'c' は連続時間系を示す.
 s1  =
 
        1
    ---------
             2
    1 + s + s
-->t=0:0.1:10;              // 解を計算する時刻：0 から 0.1 刻みで 10 まで.
-->y=csim('impuls',t,s1);   // インパルス応答の計算（s1 は厳密にプロパー）
-->plot2d(t,y)              // プロット（Matlab 書式の plot もある.）
-->clf                      // グラフの消去
-->u=ones(t);               // 単位ステップ関数
-->y=csim(u,t,s1);          // ステップ応答の計算
-->plot2d(t,y)

-->deff('[x]=u1(t)','x=sin(2*t)');  // 関数 u1 の定義
-->y=csim(u1,t,s1);        // 入力 u1 による強制応答
-->clf
-->plot2d(t,y,1)           // 色 1 で y をプロット
-->plot2d(t,u1(t),2)       // 色 2 で u1 をプロット
-->xgrid(4)                // グラフに色 4 の格子を描く
-->clf
-->xmin=0; xmax=10;
-->ymin=-2;ymax=2;
-->plot2d(t,y,1,rect=[xmin ymin xmax ymax])  // 表示範囲を指定
```

例題 1.1 種々の $\zeta \geq 0$ に対する 2 次要素

$$G(s) = \frac{1}{s^2 + 2\zeta s + 1}$$

のインパルス応答を描け.

impulse.sce

```
s=%s;
z=0:0.2:1;               // 規則的な\zeta
//z=[0.1 0.4 0.7 1 2];   // 指定した\zeta
yy=[];
t=0:0.1:20;
for i=1:length(z)
  g=1/(s^2+2*z(i)*s+1);
  gs=syslin('c',g);
  y=csim('impuls',t,gs);
  yy=[yy;y];
```

```
end
plot2d(t',yy')
xgrid(4)
```

実行

```
-->chdir('C:\scilab\control');     // impulse.sce を保存したディレクトリに変更
-->exec impulse.sce                // 実行
```

演習問題

1. 種々の $\zeta \geq 0$ に対する 2 次要素

$$G(s) = \frac{1}{s^2 + 2\zeta s + 1}$$

のステップ応答を描け.

1.5　周波数応答

ゲインと位相の計算

```
-->s=%s;
-->g1=(0.7464*s+1)/(0.2*s+1)
 g1  =
    1 + 0.7464s
    -----------
     1 + 0.2s
-->x=horner(g1,%i*5)   // s=%i5 における g1 の値. %i は虚数単位
 x  =
    2.366 + 1.366i
-->gain=abs(x)         // x の絶対値（ゲイン）
 gain  =
    2.7320161
-->phase=atan(imag(x),real(x))   // x の偏角（位相）[rad]
 phase  =
    0.5235954
-->phase*180/%pi                // ラジアンを度に変換
 ans  =
    29.999805
```

ボード線図（ゲイン特性と位相特性を同じウィンドウに描く）

bodec.sce

```
// bode diagram
s=%s;
g=1/(s+1)^4;
w=logspace(-2,2,400);      // 10 の-2 乗から 10 の 2 乗まで, 対数的に等間隔に 400 点とる.
gjw=horner(g,%i*w);
```

```
gain=abs(gjw);
gaindB=20*log10(gain);     // デシベル表示
phase=phasemag(gjw,'c');  // 位相（連続）[deg]
xsetech([0,0,1,0.5]) //xsetech([x,y,w,h]) (x,y)は図の位置，(w,h)は図の幅，高さを示す.
plot2d(w,gaindB,logflag='ln');  // ゲイン特性を片対数グラフに描く.
xgrid(4)
xsetech([0,0.5,1,0.5])
plot2d(w,phase,logflag='ln');   // 位相特性を片対数グラフに描く.
xgrid(4)
```

ボード線図（ゲイン特性と位相特性を別々のウィンドウに描く）

bodec2.sce

```
// bode diagram
s=%s;
g=1/(s+1);
w=logspace(-2,4,400);
gjw=horner(g,%i*w);
gain=abs(gjw);
gaindB=20*log10(gain);
phase=phasemag(gjw,'c');
xset('window',0);              // ウィンドウ0
plot2d(w,gaindB,logflag='ln');
xgrid(4)
xset('window',1);              // ウィンドウ1
plot2d(w,phase,logflag='ln');
xgrid(4)
```

例題 1.2 種々の $\zeta > 0$ に対する 2 次要素

$$G(s) = \frac{1}{s^2 + 2\zeta s + 1}$$

のボード線図を描け.

（解答例）

bodec3.sce

```
s=%s;
z=0.1:0.3:1;             // 規則的な\zeta
//z=[0.1 0.3 0.5 1 2];  // 指定した\zeta
w=logspace(-2,2,400);
Gjw=[];
for i=1:length(z)
  g=1/(s^2+2*z(i)*s+1);
  gjw=horner(g,%i*w);
  Gjw=[Gjw;gjw];
end
```

```
gain=abs(Gjw);
gaindB=20*log10(Gjw);
phase=phasemag(Gjw,'c');
xset('window',0);
plot2d(w',gaindB',logflag='ln');
xgrid(4)
xset('window',1);
plot2d(w',phase',logflag='ln');
xgrid(4)
```

1.6　開ループ伝達関数の特性値

ゲイン交差周波数と位相余裕

```
-->s=%s;
-->L=1/(s*(s+1));       // 開ループ伝達関数
-->sysL=syslin('c',L);
-->[pm,fr]=p_margin(sysL)
 fr  =
    0.1251199    // ゲイン交差周波数 [Hz]
 pm  =
    51.827292    // 位相余裕 [deg]
-->wp=fr*2*%pi   // [rad/s] に変換
 wp  =
    0.7861514
```

位相交差周波数とゲイン余裕

```
-->L=0.2/(s*(s^2+s+1));
-->sysL=syslin('c',L);
-->[g,fr]=g_margin(sysL)
 fr  =
    0.1591549    // 位相交差周波数 [Hz]
 g  =
    13.9794      // ゲイン余裕 [dB]
-->wq=fr*2*%pi   // [rad/s]
 wq  =
    1.
-->gm=10^(g/20)  // 絶対値に変換
 gm  =
    5.
```

演習問題　次のループ伝達関数に対するゲイン交差周波数 ω_P，位相余裕 ϕ_M，位相交差周波数 ω_Q，ゲイン余裕 G_M を求めよ．

$$L(s) = \frac{1.5}{(s+1)(s^2+s+1)}$$

（答）

$$\omega_P = 1.0379,\quad \phi_M = 39.68^\circ,\quad \omega_Q = 1.4142,\quad G_M = 2$$

1.7　閉ループ伝達関数の特性値

ゲインのピーク値

```
-->s=%s;
-->z=0.1;
-->g=1/(s^2+2*z*s+1);
-->gs=syslin('c',g);
-->fr=freson(gs)     // ピーク周波数 [Hz]（ベクトル）を求める．gs はスカラー
 fr  =
    0.1575554
-->wr=fr*2*%pi       // [rad/s] に変換
 wr  =
    0.9899495
-->Mp=abs(horner(g,%i*wr))   // ゲインのピーク値
 Mp  =
    5.0251891
-->h_norm(gs)          // gs の H 無限大ノルム
 ans  =
    5.0251891
-->z=0.1;
-->Omega_p=sqrt(1-2*z^2)     // ピーク周波数（理論式）[rad/s]
 Omega_p  =
    0.9899495
-->M_p=1/(2*z*sqrt(1-z^2))  // ゲインのピーク値（理論式）
 M_p  =
    5.0251891
```

例題 1.3 2 次要素

$$G(s) = \frac{1}{s^2 + 2\zeta s + 1}$$

の $M_p(\zeta)$ のグラフ（$\zeta = 0.1 \sim 0.7$ とする）を，freson による計算値と理論式による計算値を求めることによってそれぞれ描け．理論式は次式である．

$$M_p = \frac{1}{2\zeta\sqrt{1-\zeta^2}}$$

（解答例）

gpeaks.sce

```
// peak gains
s=%s;
z=0.1:0.05:0.7;
Mpdata=[];
```

```
for i=1:length(z)
  g=1/(s^2+2*z(i)*s+1);
  gs=syslin('c',g);
  wr=freson(gs)*2*%pi;
  Mp=abs(horner(g,%i*wr));
  Mpdata=[Mpdata Mp];
end
plot2d(z,Mpdata,2)
// theoretical
Mptinv=2*z.*sqrt(1-z.^2);
Mpt=ones(Mptinv)./Mptinv;   // 各要素の逆数
plot2d(z,Mpt,-1) // +印
```

1.8　ナイキスト線図

nyquist1.sce

```
s=%s;
g=1/(s+1);
w1=[0 0.1 0.3 0.6 1.6 3 10]; // 指定した角周波数
gjw=horner(g,%i*w1);
gre=real(gjw);
gim=imag(gjw);
plot2d(gre,gim,-2) // ×印

w=logspace(-2,2,100);
Gjw=horner(g,%i*w);
Gre=real(Gjw);
Gim=imag(Gjw);
plot2d(Gre,Gim,2)
xgrid(4)
```

1.9　根軌跡

```
-->s=%s;
-->g=1/(s*(s+1)*(s+2));
-->gs=syslin('c',g);
-->evans(gs)
-->xgrid(4)
-->clf
-->evans(gs,10)   // ゲインの上限を 10 とする.
-->xgrid(4)
```

参考文献

1. Scilab 5.2.2 Help